Combating Chart Overload with Group Control Charts

Darren Stewart, Pratt & Whitney Aircraft

Eric C. Brinsfield, Meridian Software, Inc.

ABSTRACT

Traditional control charting techniques, such as Shewhart charts, are used to track changes in one variable over time. When monitoring a manufacturing process, often more than one variable represents the state of a process at any given point in time. To monitor uniformity, a separate measurement of thickness at various locations across a manufactured unit, for example, is taken at each location. Traditional charting methods call for a separate chart for each location.

"Group charts," however, are used for plotting multiple characteristics – such as different measurement locations on each unit – on the same chart. Group charts make comparisons amongst the multiple characteristics much easier, while reducing to one the number of charts needed to monitor a particular process. Group charts can be used as exploratory tools or as formal control charts with statistical control limits. We will demonstrate how SAS® can be used to create these innovative charts.

INTRODUCTION

Processes that are characterized by a large number of parameters can present challenges to process control engineers accustomed to working with standard SPC (Statistical Process Control) tools. Traditional control charts, in general, facilitate monitoring only one process parameter per chart. In certain situations, multivariate control charts can be used to simultaneously monitor all of the parameters on the same chart. But oftentimes, SPC engineers are stuck with a potentially large number of charts to examine.

A process with 20 process parameters, for example, would require 40 control charts -- 20 to monitor the process means and another 20 to monitor the process variances. With so many charts to look at in determining the status of their production line, process engineers may experience *control chart overload*. It may take too long to examine all of the charts to make timely decisions about the process.

This paper discusses some alternative control chart methods that can illuminate problems when many parameters must be monitored. Most of the techniques described below are based on the methods published by Stephen Wise and Douglas Fair in their book Innovative Control Charting: Practical Solutions for Today's Manufacturing Environment, We used SAS software to implement their ideas.

THE PROBLEM

As an example to illustrate a situation where you might want to monitor several parameters at once, consider the production of film used in video tapes. They are constructed by taking a large roll of film, upon which a layer of magnetic solution is evenly spread. After the drying process, each large roll of film is sliced into individual tapes that are the width of the video tapes that we all use.

Because the process is performed on a wide film rather than individual strips, you may experience variation in quality across the film. We needed a way to monitor not only the stability of the entire process over time, but also the consistency across the film. In this example, we are trying to monitor optical density.

SOLUTIONS

Along with other measures of variability, we chose to utilize the **group control chart** to provide a simple visual check for across-film consistency. In general, if you need to monitor multiple parameters and some of the parameters are "similar enough", the similar measurements can be monitored together on *one* group control chart. "Similar enough" implies that, at least, the candidate process parameters must be recorded in the same units of measurement. Also, the parameters should be measuring the same basic quality characteristic, such that visualizing them together on the same chart will be useful. Wise and Fair recommend grouping no more than 5 different parameters together on the same chart. SPC engineers who implement these charts in practice can gain a better understanding of when variables should be grouped together.

In our example, we needed to take measurements of optical density at different locations across the film. As a result, the measurements were in the exact same units of measure and had the same target values. They only varied by location. Refer to Figure 1.



1 Figure 1 Location Identification on Film

THE GROUP CONTROL CHART

A **group mean (X-bar) chart** consists of two plotted points at each subgroup interval: the subgroup's maximum mean (X-bar) and the subgroup's minimum mean (X-bar).



2 Figure 2 Group X Chart

A subgroup is defined by a period of time in which sample measurements (one or more) from each of the parameters are taken. Plotted points are labeled with the appropriate location or characteristic name. A center line (Xbar bar) and/or target line can also be plotted.

To monitor the process variability, a **group range** (or **group standard deviation**) **chart** should be plotted in conjunction with the group mean chart. Group range charts consist of two plotted points at each subgroup interval: the subgroup maximum range and the subgroup minimum range. Typically, a group range chart would be used when each subgroup consists of less than 10 measurements for each parameter; group standard deviation charts are used when 10 or more points determined a subgroup.



3 Figure 3 Group R Chart

The goal of group control charts -- and any other control chart -- is to provide a warning to the process engineer that the process has either changed in some manner or is not meeting expectations. Group control charts assess the status of a process by revealing whether one or more process parameters are performing poorly *relative* to the other parameters in the group. If parameter A, for example, is producing the maximum point the majority of the time, then parameter A is most likely behaving differently than the others.

Group control charts are often used for short production runs in which there are not enough data points to establish formal control limits. By grouping similar parameters together, information from all of the groups together is more valuable than if each of the parameters were charted alone. For this reason, group control charts are ideal for short production runs. Group control charts are also very useful as informal, exploratory plotting tools.

Control limits can be included on group control charts if reliable estimates of the parameters' means and variances can be obtained. Often this is not possible for short production runs. If feasible, standard, 3-sigma limits can be plotted for both the group X and group R charts. These would be the same limits that each of the traditional control charts for individual parameters would have. One must realize, however, that the probability of a false alarm will be higher than what a standard 3-sigma chart advertises. This danger results because we are making inferences about more than one parameter at a time.

TARGET AND GROUP TARGET CHARTS

If the parameters that are grouped together have different target means, a **group target chart** must be used. Here, the observed measurements are first converted to deviations about their respective targets. By doing so, it now makes sense to group the parameters on the same chart. This chart will thus be centered around zero. If the parameters also have different target variances, the observations can be standardized completely by subtracting the target and then dividing by the target standard deviation.

CREATING GROUP CHARTS WITH SAS SOFTWARE

SAS/QC does not provide an automatic method for creating group charts. But by using base SAS and the SAS/GRAPH annotate facility in conjunction with PROC SHEWHART, you can produce the desired results. (Refer to Figure 2 and 3) You can also use GPLOT in SAS/GRAPH to create group charts, but for a consistent appearance with other control charts we produced our group control charts using the Shewhart procedure.

Step 1 Summarize the data

Summarize the data using PROC MEANS, the data step or whatever method you prefer. First you need to summarize by GROUP and LOCATION, saving the mean, range and standard deviation. Next, you resummarize by GROUP, keeping track of which location yielded the maximum and minimum values.

Step 2 Build the annotations

Using the data set that contains your minimum and maximum values, create separate annotate data sets for means, ranges and standard deviations. The following macro creates an annotate data set that will plot, label, and connect the maximum values then plot, label, and connect the minimum values. In addition, the color of the plotted point and the label will be changed to RED, if we detect that a specific location has occupied the maximum or minimum position more than expected.

```
/*-------------------------------------------------------------*
* Read the MINMAX data set to build three annotate data sets   *
* that will plot the maximum and minimum values, labels and    *
* lines. Call the macro MKANNOS three times.  The data step    *
* loops through the MINMAX data set twice, once for max values *
* and once for minimum values.                                 *
* annoname = annotate data set name                            *
* maxvar   = variable containing max value                     *
* minvar   = variable containing min value                     *
* maxlocv  = variable containing location of max value         *
* minlocv  = variable containing location of min value         *
*----------------------------------------------*/
%macro mkannos(annoname,maxvar,minvar,maxlocv,minlocv);

   data &annoname;

      length function $ 8 text $ 20 style $ 8;
      retain xsys '2' ysys '2' hsys '1';

      /*---------------------------------------*
```

```
* Read MINMAX twice, once for max and once for min.      *
*--------------------------------------*/
do annoset='UPPER','LOWER';
   do i=1 to nobs;

      set minmax point=i nobs=nobs;

      /*-----------------------------------------------*
      * X-coordinate is the group value and the          *
      * Y-coordinate is the maximum or minimum value.    *
      *-----------------------------------------------*/
      x = group;
      if annoset='UPPER' then do;
         y = &maxvar;
         if &maxlocv = "&alertmax" then
            lbcolor ='red  ';
         else
            lbcolor ='green';
      end;
      else do;
         y = &minvar;
         if &minlocv = "&alertmin" then
            lbcolor ='red  ';
         else
            lbcolor ='green';
      end;

      /*-----------------------------------------------*
      * Move to the first point in the line              *
      *-----------------------------------------------*/
      if i=1 then do;
         function= 'move';
         position= '2';
         output &annoname;
      end;

      /*-----------------------------------------------*
      * Use a symbol at each max or min value            *
      *-----------------------------------------------*/
      function= 'symbol';
      color   = lbcolor;
      style   = ' ';
      position= '2';
      size    =  4;
      text    = 'dot  ';
```

```
            output &annoname;

            /*-----------------------------------------------*
            * Connect the dots.                              *
            *-----------------------------------------------*/
            function= 'draw';
            color   = 'black';
            line    =  2;
            size    =  1;
            output &annoname;

            /*-----------------------------------------------*
            * Label the min or max with the location ID      *
            * If alert was set for a location, set color to red*
            *-----------------------------------------------*/
            function= 'label';
            color   =  lbcolor;
            size    =  6;
            position= '3';
            style   = 'Swissb';

            if annoset='UPPER' then
               text = &maxlocv;
            else
               text = &minlocv;

            output &annoname;

         end; /*---------------------- END: DO I=1 TO NOBS  */
      end; /*------------------------- END: DO UPPER, LOWER */

      STOP; /*------------------------STOP THE LOOPING      */
   run;

%mend mkannos;
```

Step 3: Create the group charts

Using data that has been summarized by GROUP, we used the HISTORY= option on the Shewhart statement to read pre-summarized data. The sample code below, shows how a combined XS group chart can be created with PROC SHEWHART.

```
title 'Group XS Chart for Optical Density';
proc shewhart history=history graphics
              annotate=meananno
              annotate2=stdanno;
  xschart optden*group / haxis=&haxmin to &haxmax by 1
                                        vaxis=&vaxminm to &vaxmaxm
```

```
                                   vaxis2=&vaxmins to &vaxmaxs
                                    nolcl noucl nolcl2 noucl2 nolimitslegend ;
run;
```

We added an extra group on each side and automatically enlarged the veritical axes in order to be sure we had room for the min and max values and labels.   The code listed above created the output displayed in Figure 4.



4 Figure 4 Group XS Chart

SUMMARY

Group charts provide an easy visual method for comparing measurements at multiple locations. Using other features within the SAS system, you could automate the monitoring of excessive maximum or minimum values. Through the SAS/GRAPH annotate facility, you can develop extremely diverse variations on the standard Shewhart chart.

Although SAS programmers can create the desired effect with Shewhart, we request that SAS Institute provide group control charts and target charts as an automatic feature in the Shewhart procedure. No only would this option provide a valuable tool, but it would reduce the chance of programmer error resulting from the complex logic required to blend summary statistics and annotate instruction.

REFERENCE INFORMATION

Darren Stewart, Quality Engineer

Pratt & Whitney Aircraft

East Hartford, Connecticutt 06108

Phone: 860-565-8807

Email: steward@pweh.com

Eric Brinsfield, President
Meridian Software, Inc.
12204 Old Creedmoor Road
Raleigh, NC 27613
Phone: 919.518.1070
FAX: 919.518.1170
email: merecb@meridian-software.com

WWW: http://www.meridian-software.com

BIBLIOGRAPHY

Wise, Stephen A. and Fair, Douglas C., 1998*, Innovative Control Charting: Practical SPC Solutions for Today's Manufacturing Environment*, Quality Press

TRADEMARKS